

Classical Cryptography

András Zsolt Sütő, Azim Kazimli, Tobias Innleggen

Saturday 22nd March, 2025

Abstract

In this project, we investigate the effectiveness of using Haskell in implementing and breaking classical ciphers. Our report focuses on three historically significant ciphers: the Caesar cipher, the Vigenère cipher, and the Playfair cipher. We develop automated methods to recognize which cipher has been used on a given ciphertext by analyzing statistical features such as letter frequencies, n -gram patterns, entropy, and periodicity. Based on the identified cipher, we then apply appropriate cryptanalytic attacks: brute-force key search for Caesar and Playfair, and a combination of the Friedman test, Kasiski examination, and frequency analysis for Vigenère. We evaluate the success rate of these attacks and the performance of our Haskell implementations. **(Final results will be inserted here upon completion, summarizing the accuracy and efficiency achieved.)**

Contents

1	Introduction	3
2	Implementing classical cryptography ciphers	3
2.1	Caesar and Keyword cipher	3
2.2	Vigenère Cipher	5
2.3	Playfair cipher	6
3	Cipher recognition	9
4	Attacks	9
4.1	Attack on Caesar and Keyword Cipher	9
4.2	Attack on Vigenère Cipher	11
4.3	Attack on Playfair Cipher	11
5	Running the program	11

6	Tests	14
7	Conclusion	14
	Bibliography	14

1 Introduction

The motivation behind this report is twofold. First, we aim to demonstrate that a functional language like Haskell can elegantly implement classical ciphers and their cryptanalysis. Second, we address a practical scenario: given an unknown piece of encrypted text, can we automatically determine the cipher used and subsequently crack it?

First we implement three classical ciphers:

- **Caesar cipher** is a simple monoalphabetic substitution cipher that shifts letters by a fixed offset; it was famously used by Julius Caesar and exemplifies the most basic substitution technique.
- **Vigenère cipher** extends this idea using a keyword to vary the shift across the message, making it a polyalphabetic cipher that was once considered "le chiffre indéchiffrable" (the indecipherable cipher) due to its resistance to simple frequency analysis.
- **Playfair cipher** is a digraph substitution cipher that encrypts pairs of letters using a 5x5 key matrix. This cypher mixes letter coordinates and thus complicates frequency patterns. [BR23]

Caesar is trivial to break with brute force, Vigenère requires analysis to uncover its repeating key, and Playfair demands strategic searching using a list of common words due to its larger key space. Both the ciphered text and the key are assumed to be in English.

We built a **cipher recognition** mechanism that inspects an unknown ciphertext and infers which cipher produced it. We base this recognition on statistical features of the ciphertext, such as letter frequency distributions, entropy and coincidence measures and periodicity, as explained in Section 3.

We measure the success rate of our attacks by testing each cipher-breaking method on sample ciphertexts and checking if the correct plaintext (or key) is recovered. We also profile the performance (time complexity and efficiency) of the Haskell solutions. Another aspect of our evaluation is the *ease of coding* and clarity of the solutions in Haskell. Throughout the project, we found that Haskell's expressive features (like higher-order functions, list comprehensions, and immutable data structures) allowed us to write cryptographic algorithms in a concise manner.

2 Implementing classical cryptography ciphers

Some more theory, some more citing like [?].

2.1 Caesar and Keyword cipher

The substitution cipher is a way of encrypting where the units of the plaintext is replaced with cipher text. There are several different kinds of substitution ciphers, but the one considered in this section is the Caesar and Keyword cipher. These ciphers are versions of the substitution cipher that works on single letters. The caesar cipher takes a int as key value, and the keyword

cipher generates a scrambled alphabet based on a keyword. To accommodate different cipher key formats in a uniform and maintainable way the typeclass `CipherInput` is used. The `CipherInput` typeclass specifies that any type used as a cipher key must implement the `create` function. This allows for treating multiple cipher types uniformly, making use of Haskell's polymorphism.

```
class CipherInput a where
    createCipherMap :: a -> M.Map Char Char
```

The caesar cipher is constructed by writing the alphabet twice, once as the normal alphabet and once with the letters shifted by a number of steps or even reversed. To create this mapping a concrete implementation of the earlier defined `CipherInput` typeclass for the input type `Int`. This instance allows integers to serve as keys for the caesar cipher. The function generates a cipher map by splitting the regular alphabet at the rotation point, and then recombining the two segments into a shifted alphabet. The shifted alphabet is then zipped with the regular alphabet, forming the mapping between the plaintext letters and the cipher letters.

```
instance CipherInput Int where
    createCipherMap :: Int -> M.Map Char Char
    createCipherMap rotation =
        let baseAlphabet = ['A'..'Z']
            (firstPart, secondPart) = splitAt (rotation `mod` 26) baseAlphabet
            flippedAlphabet = secondPart ++ firstPart
        in M.fromList $ zip baseAlphabet flippedAlphabet
```

The keyword cipher is constructed by reordering the alphabet based on a given keyword, ensuring that each letter appears only once while preserving the remaining order of the alphabet. To create this mapping, a concrete implementation of the earlier defined `CipherInput` typeclass is provided for the input type `String`. This instance allows strings to serve as keys for the keyword cipher. The function generates a cipher map by extracting unique letters from the keyword, appending the remaining letters of the alphabet in order, and then zipping this modified sequence with the regular alphabet to form the mapping between plaintext and cipher letters.

```
instance CipherInput String where
    createCipherMap :: String -> M.Map Char Char
    createCipherMap keyword =
        let baseAlphabet = ['A'..'Z']
            uniqueKeyword = map toUpper . nub $ keyword
            remainingLetters = baseAlphabet \\ uniqueKeyword
            cipherAlphabet = uniqueKeyword ++ remainingLetters
        in M.fromList $ zip baseAlphabet cipherAlphabet
```

The `generateCipherTextFromMap` function transforms plaintext using a cipher mapping by filtering out non-alphabetic characters and replacing valid ones through `M.lookup`, which returns a `Maybe Char`. The use of `mapMaybe` ensures only successful lookups (`Just` values) are kept, while non-alphabetic or unmapped characters are discarded. The encrypted output is then formatted into five-character chunks omitting punctuation and spaces using a recursive `chunk` function. This is done in order to disguise word boundaries from the plaintext as well as to help avoid transmission errors.

The `encryptCaesarAndKeyword` function orchestrates this by generating the cipher map with `createCipherMap` and passing it to `generateCipherTextFromMap`. Haskell's `Maybe` type makes this implementation safe by forcing explicit handling of missing values, preventing common lookup errors found in imperative languages.

```

generateCipherTextFromMap :: String -> M.Map Char Char -> String
generateCipherTextFromMap plainTxt pairs =
    let filteredText = mapMaybe (\c -> if isAlpha c then M.lookup (toUpper c) pairs else
        Nothing) plainTxt
    in unwords (chunk 5 filteredText)

chunk :: Int -> String -> [String]
chunk _ [] = []
chunk n str = take n str : chunk n (drop n str)

encryptCaesarAndKeyword :: CipherInput p => p -> String -> String
encryptCaesarAndKeyword key plaintext =
    let cipherMap = createCipherMap key
    in generateCipherTextFromMap plaintext cipherMap

```

The decryption process of both the caesar and keyword cipher is straightforward, as it involves inverting the cipher map used for encryption. This is exactly what happens in the functions below. The main decryption function, `decryptCaesarAndKeyword`, first generates the key map using `createCipherMap`, then inverts it using `invertCipherMap`, which swaps the keys and values by converting the map to a list of pairs with `M.toList`, mapping each `(k, v)` pair to `(v, k)`, and reconstructing a new map with `M.fromList`. This functional approach is both concise and efficient because it avoids explicit loops and operates in a single pass over the map structure. The `decryptCipherText` function then applies the decryption map while preserving spaces and punctuation, using `map` to iterate over the text and `fromMaybe` to handle unmapped characters safely. This ensures that decryption is expressive and declarative, leveraging Haskell's built-in higher-order functions and persistent data structures to process transformations in a clear and optimized manner without unnecessary state mutations.

```

-- Invert a cipher map to create a decryption map
invertCipherMap :: M.Map Char Char -> M.Map Char Char
invertCipherMap = M.fromList . map (\(k, v) -> (v, k)) . M.toList

-- Process ciphertext for decryption (preserve spaces and punctuation)
decryptCipherText :: String -> M.Map Char Char -> String
decryptCipherText cipherTxt decryptMap =
    map (\c -> if isAlpha c
        then fromMaybe c (M.lookup (toLower c) decryptMap)
        else c) cipherTxt

-- Main decryption function that works with both String and Int keys
decryptCaesarAndKeyword :: CipherInput p => p -> String -> String
decryptCaesarAndKeyword key ciphertext =
    let cipherMap = createCipherMap key
        decryptMap = invertCipherMap cipherMap
    in decryptCipherText ciphertext decryptMap

```

2.2 Vigenère Cipher

```

module Vigenere where

```

Unlike a simple monoalphabetic shift, the Vigenère cipher uses multiple shifts determined by a repeating keyword. In Haskell, this design naturally translates into a few elegant higher-order functions, list manipulations, and local bindings.

We define `vigenere`, which takes a higher order function `op`, a `key` and a plaintext, and returns the ciphertext. We rely on `cycle` to stream each character of the `key` infinitely, then pair it

with each plaintext character. The recursive helper `enc` uses pattern matching to either shift a character or leave them as-is, depending on whether it is an alphabetical letter - this is an elegant way of replacing manual loop constructs.

```
vigenere :: (Int -> Int -> Int) -> String -> String -> String
vigenere op key = enc (cycle key)
  where
    enc _ [] = []
    enc [] _ = []
    enc (k:ks) (t:ts)
      | isAlpha t = shift k t : enc ks ts
      | otherwise = t : enc (k:ks) ts
```

Here, `shift` calculates the offset by converting the key character `k` to uppercase and subtracting 'A'. We conditionally decide the base code point (uppercase or lowercase) for the plaintext letter and then wrap around with modular arithmetic. Because all values are passed immutably, we avoid side effects and can directly transform each character.

```
shift k t =
  let base = if isUpper t then ord 'A' else ord 'a'
      offset = ord (toUpper k) - ord 'A'
  in chr $ base + mod ((ord t - base) `op` offset) 26
```

By passing an operator (+ for encryption or - for decryption) we generalized our cipher logic into one function, therefore encoding and decoding are defined as specializations of `vigenere` with the appropriate operator:

```
vigenereEncode :: String -> String -> String
vigenereEncode key text = vigenere (+) key text

vigenereDecode :: String -> String -> String
vigenereDecode key text = vigenere (-) key text
```

Since each transformation is defined as a pure function, data flows in one direction without side effects. The pattern matching in `enc` clarifies when to apply shifts (only on alphabetic characters) and when to keep a character intact.

2.3 Playfair cipher

In our Haskell implementation of the Playfair cipher, we express the algorithm in a functional style, highlighting Haskell's strengths in handling complex string manipulations and modular code design.

The core components of our implementation include:

- **Table Construction:** We generate a 5 x 5 table from a user-supplied keyword by removing duplicate letters and appending the remaining letters of the alphabet (with 'J' treated as 'I'). This process is performed by the `createTable` function.
- **Text Preparation:** The plaintext is filtered to remove non-letter characters, converted to uppercase, and normalized (replacing 'J' with 'I') using the `prepareText` function.
- **Digram Generation:** The normalized text is split into pairs (digrams) with the

`makeDigrams` function. When duplicate letters occur within a pair, a filler (typically 'X') is inserted to ensure proper encryption.

- **Encryption and Decryption:** Depending on whether a pair of letters is in the same row, same column, or forms the corners of a rectangle in the table, we perform the appropriate substitution. This logic is implemented in the `encryptDigram` and `decryptDigram` functions, while the overall encryption and decryption are handled by the `encryptPlayfair` and `decryptPlayfair` functions.

We begin with the module declaration:

```
module Playfair where
```

Below are the key functions used in our implementation:

```
import Data.Char (toUpper, isAlpha)
import Data.List (nub)

-- The alphabet for the cipher (note that 'J' is omitted)
alphabet :: [Char]
alphabet = filter (/= 'J') ['A'..'Z']
```

Table Construction: The `createTable` function builds the 5 x 5 table by first cleaning the keyword and then appending any missing letters.

```
createTable :: String -> [Char]
createTable keyword =
  let cleaned    = map (\c -> if c == 'J' then 'I' else c)
                    $ filter isAlpha $ map toUpper keyword
      keyUnique  = nub cleaned
  in keyUnique ++ filter ('notElem' keyUnique) alphabet
```

Text Preparation: The `prepareText` function filters and normalizes the input text.

```
prepareText :: String -> String
prepareText = map (\c -> if c == 'J' then 'I' else c)
              . filter isAlpha . map toUpper
```

Digram Generation: The `makeDigrams` function splits the text into pairs (digrams), inserting an 'X' when two consecutive letters are identical or when padding is required.

```
makeDigrams :: String -> [String]
makeDigrams []          = []
makeDigrams [x]         = [[x, 'X']]
makeDigrams (x:y:xs)
  | x == y    = [x, 'X'] : makeDigrams (y:xs)
  | otherwise = [x, y]   : makeDigrams xs
```

Finding Character Positions: The helper function `chunksOf` is used to break lists into chunks and `findPosition` locates a character within the 5×5 table.

```
chunksOf :: Int -> [a] -> [[a]]
chunksOf _ [] = []
chunksOf n xs = take n xs : chunksOf n (drop n xs)
```

```

findPosition :: [[Char]] -> Char -> (Int, Int)
findPosition table c =
  let positions = [ (r, col)
                    | (r, row) <- zip [0..] table
                      , (col, ch) <- zip [0..] row
                      , ch == c ]
  in case positions of
    [] -> error ("Character not found: " ++ [c])
    (p:_) -> p

```

Encryption and Decryption: The functions `encryptDigram` and `decryptDigram` perform the letter substitutions based on the rules of the Playfair cipher.

```

encryptDigram :: [[Char]] -> String -> String
encryptDigram table [a,b] =
  let (rowA, colA) = findPosition table a
      (rowB, colB) = findPosition table b
  in if rowA == rowB then
    -- Same row: shift right
    [ table !! rowA !! ((colA + 1) `mod` 5)
    , table !! rowB !! ((colB + 1) `mod` 5) ]
  else if colA == colB then
    -- Same column: shift down
    [ table !! ((rowA + 1) `mod` 5) !! colA
    , table !! ((rowB + 1) `mod` 5) !! colB ]
  else
    -- Rectangle: swap columns
    [ table !! rowA !! colB
    , table !! rowB !! colA ]
encryptDigram _ _ = error "Invalid digram length"

decryptDigram :: [[Char]] -> String -> String
decryptDigram table [a,b] =
  let (rowA, colA) = findPosition table a
      (rowB, colB) = findPosition table b
  in if rowA == rowB then
    -- Same row: shift left
    [ table !! rowA !! ((colA - 1) `mod` 5)
    , table !! rowB !! ((colB - 1) `mod` 5) ]
  else if colA == colB then
    -- Same column: shift up
    [ table !! ((rowA - 1) `mod` 5) !! colA
    , table !! ((rowB - 1) `mod` 5) !! colB ]
  else
    -- Rectangle: swap columns
    [ table !! rowA !! colB
    , table !! rowB !! colA ]
decryptDigram _ _ = error "Invalid digram length"

```

Finally, the overall `encryptPlayfair` and `decryptPlayfair` functions tie these components together:

```

encryptPlayfair :: String -> String -> String
encryptPlayfair keyword text =
  let tableFlat = createTable keyword
      table      = chunksOf 5 tableFlat
      prepared   = prepareText text
      digrams    = makeDigrams prepared
  in concatMap (encryptDigram table) digrams

decryptPlayfair :: String -> String -> String
decryptPlayfair keyword text =
  let tableFlat = createTable keyword
      table      = chunksOf 5 tableFlat
      digrams    = chunksOf 2 (prepareText text)
  in concatMap (decryptDigram table) digrams

```


In summary, our Playfair cipher implementation in Haskell clearly demonstrates how breaking down a problem into small, composable functions can result in clean, modular, and maintainable code. By leveraging Haskell's strong type system and functional abstractions, we can model classical encryption algorithms in an elegant and concise manner.

3 Cipher recognition

This section is about cypher recognition.

4 Attacks

This section will have the implementation for the attacks on the ciphers.

4.1 Attack on Caesar and Keyword Cipher

The Caesar cipher has a key size of only one, making it vulnerable to a brute force attack. This approach involves iterating through all possible keys (1-26) and decrypting the ciphertext with each key. The function returns a list of tuples, where the first element is the key and the second element is the corresponding plaintext. This exhaustive approach is feasible due to the limited keyspace of the Caesar cipher.

```
bruteforceSubstitution :: Int -> String -> [(Int, String)]
bruteforceSubstitution iteration ciphertext
  | iteration > 26 = []
  | otherwise =
    let plaintext = decryptCaesarAndKeyword iteration ciphertext
    in (iteration, plaintext) : bruteforceSubstitution (iteration + 1) ciphertext

bruteforceCaesar :: String -> [(Int, String)]
bruteforceCaesar = bruteforceSubstitution 1
```

The keyword cipher is more complex to break, as the key size is variable. In order to break this cipher the fact that the properties of the natural language is preserved in the ciphertext of the keyword cipher. In any given stretch of written language, certain letters and combinations of letters occur with differing frequencies. In addition to this, a characteristic distribution of letter frequency in the given language will be almost the same for every sample of that language. These properties can be exploited to reconstruct the alphabet used for encryption.

The `letterFrequencies` function counts the occurrences of each letter in the input text and returns a map where the keys are the letters and the values are the counts. These values can be used to find the most common letters in the ciphertext, and map these to the most common letters in the English language.

```
-- Basic constants and alphabet
alphabet :: [Char]
alphabet = ['a'..'z']
```

```

-- Pad a character list with remaining alphabet letters
padToFullAlphabet :: [Char] -> [Char]
padToFullAlphabet chars = chars ++ (alphabet \\ chars)

-- Generate initial mapping based on frequency analysis
generateInitialCrackMap :: String -> M.Map Char Char
generateInitialCrackMap text =
  M.fromList $ zip sortedFreqChars frequencyAlphabet
  where
    freqMap = letterFrequencies text
    sortedFreqChars = padToFullAlphabet $ sortedChars $ sortedPairs freqMap
    frequencyAlphabet = "etaoinshrdlucmfpgwybvqxjz" -- Standard English frequency order

-- Score text using trigram model - optimized version
scoreTrigrams :: M.Map String Double -> String -> Double
scoreTrigrams trigramMap text =
  -- Sum up found trigrams, with a default penalty for unknown trigrams
  sum [M.findWithDefault defaultPenalty trigram trigramMap |
    i <- [0 .. length cleaned - 3],
    let trigram = take 3 (drop i cleaned)]
  where
    cleaned = normalizeText text
    -- Default penalty for unknown trigrams (should be less than minimum log probability)
    defaultPenalty = -20.0 -- Adjust this value based on your dataset

-- Swap two random letters in the key map
swapTwoLetters :: M.Map Char Char -> IO (M.Map Char Char)
swapTwoLetters m = do
  a <- randomRIO ('a', 'z')
  b <- randomRIO ('a', 'z')
  if a == b
    then swapTwoLetters m -- retry until different
    else do
      -- Find which cipher letters map to a and b
      let revMap = M.fromList [(v, k) | (k, v) <- M.toList m]
      keyA = M.findWithDefault a a revMap
      keyB = M.findWithDefault b b revMap
      -- Swap the mappings
      return $ M.insert keyA b $ M.insert keyB a m

-- Hill climbing optimization with early stopping and temperature control
hillClimb :: Int -> String -> M.Map String Double -> M.Map Char Char ->
  IO (M.Map Char Char, String, Double)
hillClimb maxIter cipher logTrigrams initialKey = do
  let initialPlain = decryptCipherText cipher initialKey
      initialScore = scoreTrigrams logTrigrams initialPlain

  -- Use a helper function with an accumulator to track best result
  go maxIter initialKey initialPlain initialScore initialKey initialPlain initialScore 0
  where
    go 0 _ _ _ bestKey bestPlain bestScore _ =
      return (bestKey, bestPlain, bestScore)

    go n currKey currPlain currScore bestKey bestPlain bestScore unchanged
      -- Early stopping if no improvement for many iterations
      | unchanged > 1000 = return (bestKey, bestPlain, bestScore)
      | otherwise = do
        -- Try a new key
        newKey <- swapTwoLetters currKey
        let newPlain = decryptCipherText cipher newKey
            newScore = scoreTrigrams logTrigrams newPlain

        -- Determine whether to accept the new key
        if newScore > currScore
          then
            -- Accept improvement
            let (newBestKey, newBestPlain, newBestScore, newUnchanged) =
                if newScore > bestScore
                  then (newKey, newPlain, newScore, 0) -- New best found
                  else (bestKey, bestPlain, bestScore, unchanged + 1)
                in go (n-1) newKey newPlain newScore newBestKey newBestPlain
                    newBestScore newUnchanged

```

```

        else
            -- Reject the change
            go (n-1) currKey currPlain currScore bestKey bestPlain bestScore (
                unchanged + 1)

-- Process a text file and build a trigram model efficiently
buildTrigramModel :: FilePath -> IO (M.Map String Double)
buildTrigramModel filePath = do
    content <- readFile filePath
    -- Process the file in smaller chunks to avoid memory issues
    let normalized = normalizeText content
        -- Process in chunks of reasonable size
        chunks = chunksOf 10000 normalized
        -- Count trigrams in each chunk
        chunkCounts = map (countNgrams 3) chunks
        -- Combine all counts
        combined = foldl (M.unionWith (+)) M.empty chunkCounts
    -- Convert to log probabilities
    return $ logProbabilities $ mapProbabilities combined
where
    chunksOf _ [] = []
    chunksOf n xs = take n xs : chunksOf n (drop n xs)

-- Main function
main :: IO ()
main = do
    putStrLn "Loading language model from War and Peace..."
    trigramProbs <- buildTrigramModel "lib/WarAndPeace.txt"

    putStrLn "Reading cipher text..."
    ciphertext <- readFile "lib/cipher.txt"
    let normalized = normalizeText ciphertext

    putStrLn "Generating initial mapping based on frequency analysis..."
    let initKey = generateInitialCrackMap normalized

    putStrLn "Starting hill climbing optimization..."
    (bestKey, decrypted, score) <- hillClimb 1000000 normalized trigramProbs initKey

    putStrLn "\nBest Decryption:"
    putStrLn decrypted
    putStrLn $ "\nScore: " ++ show score

    -- Print the mapping for analysis
    putStrLn "\nMapping (cipher -> plain):"
    let sortedMapping = sortBy (comparing fst) $ M.toList bestKey
    mapM_ \(c, p) -> putStrLn $ [c] ++ " -> " ++ [p]) sortedMapping

```

4.2 Attack on Vigenère Cipher

4.3 Attack on Playfair Cipher

5 Running the program

This section (our main file) provides an interface for encoding and decoding messages using different cipher techniques, including Vigenère, Substitution, and Playfair ciphers. The program will prompt the user to enter a key and a message, then display the encoded and decoded results for each cipher method.

```

{-# LANGUAGE OverloadedStrings #-}

module Main where

```

```

import Vignere
import CaesarAndKeywordCipher
import Playfair

import System.IO (hFlush, stdout)
import Data.Char (toLower)
import Data.List (isInfixOf)

-- Assume these functions are defined in the corresponding cipher breaker modules.
-- For the purpose of a self-contained example, we provide dummy implementations.
caesarAttack :: String -> IO [AttackStat]
caesarAttack ciphertext = return
  [ AttackStat "Caesar Attack" "dummy-caesar-key"
    ("Decrypted (Caesar): " ++ ciphertext) 0.12 90.0 85.0 ]

vignereAttack :: String -> IO [AttackStat]
vignereAttack ciphertext = return
  [ AttackStat "Vignere Attack" "dummy-vignere-key"
    ("Decrypted (Vignere): " ++ ciphertext) 0.25 80.0 75.0 ]

playfairAttack :: String -> IO [AttackStat]
playfairAttack ciphertext = return
  [ AttackStat "Playfair Attack" "dummy-playfair-key"
    ("Decrypted (Playfair): " ++ ciphertext) 0.35 70.0 65.0 ]

-----
-- Types

-- | CipherType represents the type of cipher.
data CipherType = Caesar | Vignere | Playfair
  deriving (Show, Eq)

-- | AttackStat holds the results from an attack.
data AttackStat = AttackStat {
  methodName    :: String,
  key           :: String,
  decipheredText :: String,
  runtime       :: Float,
  keyMatch      :: Float,
  textMatch     :: Float
} deriving (Show, Eq)

-----
-- Cipher Recognition

-- | Recognizes the cipher type from the input ciphertext.
cipherRecognizer :: String -> IO (Maybe CipherType)
cipherRecognizer input =
  let lowerInput = map toLower input
  in if "caesar" `isInfixOf` lowerInput
    then return (Just Caesar)
    else if "vignere" `isInfixOf` lowerInput
    then return (Just Vignere)
    else if "playfair" `isInfixOf` lowerInput
    then return (Just Playfair)
    else do
      putStrLn "Cipher could not be recognized. (maybe some details)"
      return Nothing

-----
-- Attack Result Printer

-- | Pretty prints a list of AttackStat.
attackStatPrinter :: [AttackStat] -> IO ()
attackStatPrinter stats = mapM_ printStat stats
  where
    printStat stat = do
      putStrLn $ "Name of attack: " ++ methodName stat
      putStrLn $ "Key: " ++ key stat
      putStrLn $ "Deciphered text: " ++ decipheredText stat
      putStrLn $ "Runtime: " ++ show (runtime stat) ++ " seconds"
      putStrLn $ "Key match: " ++ show (keyMatch stat) ++ "%"

```

```

        putStrLn $ "Text match: " ++ show (textMatch stat) ++ "%"
        putStrLn ""

-----

-- Global Attack Function

runAttack :: (String -> IO [AttackStat]) -> String -> IO (Maybe [AttackStat])
runAttack attackFunc ciphertext = do
    stats <- attackFunc ciphertext
    attackStatPrinter stats
    return (Just stats)

-- | Executes the appropriate attack based on the recognized cipher.
attack :: Maybe CipherType -> String -> IO (Maybe [AttackStat])
attack Nothing _ = return Nothing
attack (Just cipherType) ciphertext =
    case cipherType of
        Caesar    -> runAttack caesarAttack ciphertext
        Vignere    -> runAttack vignereAttack ciphertext
        Playfair   -> runAttack playfairAttack ciphertext

-----

-- Encrypt/Decrypt menu

encryptDecryptMenu :: IO ()
encryptDecryptMenu = do
    putStrLn "\n=== Encrypt / Decrypt Ciphers ==="
    putStrLn "Select the cipher:"
    putStrLn "1. Caesar"
    putStrLn "2. Vigenere"
    putStrLn "3. Playfair"
    putStr "Your choice: "
    hFlush stdout
    cipherChoice <- getLine

    putStrLn "\nEnter the key:"
    putStr "Key: "
    hFlush stdout
    key <- getLine

    putStrLn "\nEnter the message:"
    putStr "Message: "
    hFlush stdout
    message <- getLine

    case cipherChoice of
        "1" -> do
            let encoded = encryptCaesarAndKeyword key message
            let decoded = decryptCaesarAndKeyword key encoded
            putStrLn "\n--- Caesar Cipher ---"
            putStrLn $ "Encoded: " ++ encoded
            putStrLn $ "Decoded: " ++ decoded
        "2" -> do
            let encoded = vigenereEncode key message
            let decoded = vigenereDecode key encoded
            putStrLn "\n--- Vigenere Cipher ---"
            putStrLn $ "Encoded: " ++ encoded
            putStrLn $ "Decoded: " ++ decoded
        "3" -> do
            let encoded = encryptPlayfair key message
            let decoded = decryptPlayfair key encoded
            putStrLn "\n--- Playfair Cipher ---"
            putStrLn $ "Encoded: " ++ encoded
            putStrLn $ "Decoded: " ++ decoded
        _ -> putStrLn "Invalid cipher selection."

-- Recognize attack menu
recognizeAttackMenu :: IO ()
recognizeAttackMenu = do
    putStrLn "\n=== Recognize and Attack Cipher ==="
    putStr "Enter ciphertext: "
    hFlush stdout

```

```

ciphertext <- getLine
mCipher <- cipherRecognizer ciphertext
_ <- attack mCipher ciphertext
return ()

-----

-- Main Menu

main :: IO ()
main = do
  putStrLn "=== Main Menu ==="
  putStrLn "1. Encrypt / Decrypt text"
  putStrLn "2. Recognize and attack cipher"
  putStr "Select an option (1 or 2): "
  hFlush stdout
  option <- getLine
  case option of
    "1" -> encryptDecryptMenu
    "2" -> recognizeAttackMenu
    _   -> putStrLn "Invalid option selected."

```

6 Tests

We now use the library QuickCheck to randomly generate input for the ciphers, and test whether they work correctly. We can also use QuickCheck to test attacks maybe?

7 Conclusion

Here we conclude what has to be concluded.

References

- [BR23] R. Banoth and R. Regar. Classical and modern cryptography for beginners. *Springer*, 2023.